

## Dodatek B — Deklaracje elementów.

Poniżej zestawione zostały w kolejności alfabetycznej deklaracje elementów dostępne w programie PSpice. Każda z nich uzupełniona jest o przykład użycia i komentarz.

---

### **B**                      **TRANZYSTOR POLOWY ZŁĄCZOWY GAAS**

Postać ogólna:

B..... \_dren \_bramka \_źródło \_nazwa\_modelu [\_area]

Przykłady:

BIN 100 1 0 GFAST

B13 22 14 23 GNOM 2.0

Tranzystor polowy złączowy GaAs opisywany jest modelem matematycznym zbliżonym do modelu stosowanego dla zwykłego tranzystora polowego złączowego wykonanego w krzemie. Liczba podana w polu **\_area** określa ile razy pole powierzchni zajmowane przez deklarowany przyrząd jest większe od pola powierzchni założonego w modelu.

Parametry: strona 160.

Opis modelu: strony 159, 160.

---

### **C**                                      **KONDENSATOR**

Postać ogólna:

C..... n+ n- [\_nazwa\_modelu] \_pojemność [IC=\_v]

Przykłady:

C1 10 0 0.1U

C2 1 2 0.2E-12 IC=1.5V

CFDBK 3 33 CMOD 10pF

Dodatni biegun kondensatora to **n+**, ujemny biegun to **n-**. Dodatni prąd płynie do bieguna dodatniego do bieguna ujemnego przez kondensator. Wartość pojemności podana w polu **\_pojemność** może być dodatnia lub ujemna nie może być jednak równa zero. Za pomocą modelu opisać można zależność pojemności od temperatury i napięcia panującego na zaciskach kondensatora. W polu **\_v** podaje się wartość napięcia panującego na zaciskach deklarowanego przyrządu w chwili rozpoczęcia analizy stanu nieustalonego.

Parametry: strona 114.

Opis modelu: strona 114.

**D****DIODA**

Postać ogólna:

D..... n+ n- \_nazwa\_modelu [\_area]

Przykłady:

DCLAMP 14 0 DMOD

D13 15 17 SWITCH 1.5

Anoda diody to **n+**, katoda diody to **n-**. Dodatni prąd płynie od anody do katody przez diodę. Liczba podana w polu **\_area** określa ile razy pole powierzchni zajmowane przez deklarowany przyrząd jest większe od pola powierzchni założonego w modelu.

Parametry: strona 130.

Opis modelu: strona 129.

**E****ŹRÓDŁO PRĄDU STEROWANE NAPIĘCIEM**

Postać ogólna:

E..... n+ n- nc+ nc- \_wzmocnienie

E..... n+ n- POLY(\_rząd) <lista\_par\_węzłów> <lista\_współczynników>

Przykłady:

EBUFF 1 2 10 11 1.0 ;źródło liniowe

EAMP 13 0 POLY(1) 26 0 500 ;źródło liniowe!

ENONL 1 11 POLY(2) 3 0 4 0 0 3 .2 0.005 ;źródło nieliniowe

Dodatni biegun źródła to **n+**, ujemny biegun źródła to **n-**. Dodatni prąd płynie od bieguna dodatniego do ujemnego przez źródło. Napięcie sterujące to różnica potencjałów między węzłem o numerze **nc+** i węzłem o numerze **nc-**. Wzmocnienie źródła podaje się w polu **\_wzmocnienie**. W przypadku źródła nieliniowego (słowo kluczowe **POLY**) w polu **\_rząd** podaje się liczbę napięć wpływających na wartość napięcia generowanego przez źródło sterowane na zaciskach **n+,n-**. W tym przypadku na liście węzłów sterujących **\_lista\_par\_węzłów** trzeba umieścić liczbę par węzłów (każda para to jedno napięcie sterujące) równą liczbie podanej w polu **\_rząd**. Współczynniki wielomianu opisującego związek między napięciem sterowanym i napięciami sterującymi podane są na liście **\_lista\_współczynników**. Współczynniki wielomianu: strona 88.

**F****ŹRÓDŁO PRĄDU STEROWANE PRĄDEM**

Postać ogólna:

F..... n+ n- \_nazwa\_SEM \_wzmocnienie

F..... n+ n- POLY(\_rząd) <lista\_nazw> <lista\_wsp>

Przykłady:

FSENSE 1 2 VSENSE 10.0 ;źródło liniowe

FAMP 13 0 POLY(1) VIN 500 ;źródło liniowe!

FNONL 1 11 POLY(2) V1 V2 0 13.6 0.2 .005 ;źródło nieliniowe

Dodatni biegun źródła to **n+**, ujemny biegun źródła to **n-**. Dodatni prąd płynie od bieguna dodatniego do ujemnego przez źródło. Prąd sterujący płynie przez niezależne źródło napięcia o nazwie podanej w polu **\_nazwa\_SEM**. Wzmocnienie źródła podaje się w polu **\_wzmoc-**

**nienie.** W przypadku źródła nieliniowego (słowo kluczowe **POLY**) w polu **\_rząd** podaje się liczbę prądów wpływających na wartość prądu źródła sterowanego. W tym przypadku na liście nazw niezależnych źródeł napięcia **\_lista\_nazw** trzeba umieścić nazwy źródeł, przez które płyną prądy sterujące — liczba nazw musi być równa liczbie podanej w polu **\_rząd**. Współczynniki wielomianu opisującego związek między sterowanym prądem i prądami sterującymi podane są na liście **\_lista\_wsp**.  
Współczynniki wielomianu: strona 88.

## **G**                    **ŹRÓDŁO PRĄDU STEROWANE NAPIĘCIEM**

Postać ogólna:

G..... n+ n- nc+ nc-    \_transkonduktancja  
G..... n+ n- POLY(\_rząd) <\_lista\_par\_węzłów>    <\_lista\_wsp>

Przykłady:

GBUFF    1 2      10 11 1.0                            ;źródło liniowe  
GAMP     13 0    POLY(1) 26 0    500                            ;źródło liniowe!  
GNONL    1 11    POLY(2) 3 0 4 0    0 13 0.2 .005                ;źródło nieliniowe

Dodatni biegun źródła to **n+**, ujemny biegun źródła to **n-**. Dodatni prąd płynie od bieguna dodatniego do ujemnego przez źródło. Napięcie sterujące to różnica potencjałów między węzłem o numerze **nc+** i węzłem o numerze **nc-**. Transkonduktancję źródła podaje się w polu **\_transkonduktancja**. W przypadku źródła nieliniowego (słowo kluczowe **POLY**) w polu **\_rząd** podaje się liczbę napięć wpływających na wartość prądu źródła sterowanego. W tym przypadku na liście węzłów sterujących **\_lista\_par\_węzłów** trzeba umieścić liczbę par węzłów (każda para to jedno napięcie sterujące) równą liczbie podanej w polu **\_rząd**. Współczynniki wielomianu opisującego związek między sterowanym prądem i napięciami sterującymi podane są na liście **\_lista\_wsp**.

Współczynniki wielomianu: strona 88.

## **H**                    **ŹRÓDŁO NAPIĘCIA STEROWANE PRĄDEM**

Postać ogólna:

H.....    n+ n-      \_nazwa\_SEM      \_transrezystancja  
H.....    n+ n-      POLY(\_rząd)    <\_lista\_nazw>    <\_lista\_wsp>

Przykłady:

HSENSE   1 2      VSENSE   10.0                            ;źródło liniowe  
HAMP     13 0    POLY(1)   VIN    500                            ;źródło liniowe!  
HNONL    1 11    POLY(2)   V1 V2    0 13.6 0.2    .005                ;źródło nieliniowe

Dodatni biegun źródła to **n+**, ujemny biegun źródła to **n-**. Dodatni prąd płynie od bieguna dodatniego do ujemnego przez źródło. Prąd sterujący płynie przez niezależne źródło napięcia o nazwie podanej w polu **\_nazwa\_SEM**. Transrezystancję źródła podaje się w polu **\_transrezystancja**. W przypadku źródła nieliniowego (słowo kluczowe **POLY**) w polu **\_rząd** podaje się liczbę prądów wpływających na wartość prądu źródła sterowanego. W tym przypadku na liście nazw niezależnych źródeł napięcia **\_lista\_nazw** trzeba umieścić nazwy źródeł, przez które płyną prądy sterujące — liczba nazw musi być równa liczbie podanej w polu **\_rząd**. Współczynniki wielomianu opisującego związek między sterowanym napięciem

i prądami sterującymi podane są na liście **\_lista\_wsp**.

Współczynniki wielomianu: strona 88.

## I NIEZALEŻNE ŹRÓDŁO PRĄDU

Postać ogólna:

I..... n+ n- [DC \_wdc] [AC \_mod [\_faz]] [\_przebieg\_czasowy]

Przykłady:

IBIAS 13 0 2.3mA

IAC 2 3 AC .001

IACF 2 3 AC .001 37

IPUL 2 3 PULSE(0mA 1mA)

IV1 3 2 DC 1 AC 2 90 SIN(0.2mA 1mA 1.5kHz)

Dodatni biegun źródła to **n+**, ujemny biegun źródła to **n-**. Dodatni prąd płynie od bieguna dodatniego do ujemnego przez źródło. Składową stałą prądu źródła podaje się w polu **\_wdc** po słowie kluczowym **DC**. Po słowie kluczowym **AC** podaje się amplitudę (pole **\_mod**) i fazę (pole **\_faz**) składowej zmiennej źródła. Na końcu deklaracji można podać przebieg czasowy prądu źródła. Możliwe są następujące przebiegi:

**EXP** Ekspotencjalny — opis na stronie 76.

**PULSE** Przebieg prostokątny — opis na stronie 74.

**PWL** Odcinkowo–liniowy — opis na stronie 77.

**SFFM** Sinusoida o częstotliwości modulowanej sinusoidą — opis na stronie 77.

**SIN** Sinusoidalny — opis na stronie 75.

Patrz strony: 9, 46, 73.

## J TRANZYSTOR POŁOWY ZŁĄCZOWY

Postać ogólna:

J..... \_dren \_bramka \_źródło \_nazwa\_modelu [\_area]

Przykłady:

JIN 100 1 0 JFAST

J23 22 14 23 JNOM 2.0

Liczba podana w polu **\_area** określa ile razy pole powierzchni zajmowane przez deklarowany przyrząd jest większe od pola powierzchni założonego w modelu.

Parametry: strona 155.

Opis modelu: strona 156.

## K SPRZĘŻENIE CEWEK (NIELINIOWY RDZEŃ MAGNETYCZNY)

Postać ogólna:

K..... <\_lista\_nazw\_IND> \_wsp\_sprzężenia

K..... <\_lista\_nazw\_IND> \_wsp\_sprzężenia \_nazwa\_modelu

Przykłady:

K1 L12 LWZ 0.82

KTRAF L1 L2 0.99

KNONL L1 L2 L3 L4 0.98 F3001

Element sprzęga magnetycznie cewki, których nazwy wymieniono na liście **\_lista\_nazw\_IND**. Bezwymiarowy współczynnik sprzężenia podaje się w polu **\_wsp\_sprzeżenia**. Jeżeli podana zostanie nazwa modelu (pole **\_nazwa\_modelu**) to sprzężenie magnetyczne realizowane jest poprzez nieliniowy rdzeń magnetyczny. W tym przypadku należy pamiętać, że:

- W deklaracjach sprzęganych cewek zamiast indukcyjności podaje się liczbę zwojów.
- Na liście **\_lista\_nazw\_IND** może pojawić się tylko jedna nazwa cewki. W ten sposób można modelować cewkę z nieliniowym rdzeniem magnetycznym.
- Materiał magnetyczny opisywany jest zmodyfikowanym modelem Jiles–a Atherton–a.

Parametry: strona 117.

Opis modelu: strona 118.

**L****CEWKA MAGNETYCZNA**

Postać ogólna:

L..... n+ n- [\_nazwa\_modelu] \_indukcyjność [IC=\_i]

Przykłady:

```
LLOAD 15 0 20mH
L2 1 2 1.2E-6
LGEN 3 42 LMODE 0.03
LSENSE 5 12 2UH IC=2mA
```

Dodatni biegun cewki to **n+**, ujemny biegun cewki to **n-**. Dodatni prąd płynie od bieguna dodatniego do ujemnego przez cewkę. Cewka może być opisywana modelem, którego nazwę podaje się w polu **\_nazwa\_modelu**. Model pozwala uwzględnić zmiany indukcyjności z temperaturą oraz prostych zjawisk nieliniowych. Jednak aby dokładnie opisać zachowanie cewki z nieliniowym rdzeniem magnetycznym konieczne jest zadeklarowanie elementu typu **K.....** — sprzężenia cewek. Wartość indukcyjności podana w polu **\_indukcyjność** może być dodatnia lub ujemna nie może być jednak równa zero. W polu **\_i** podaje się wartość prądu płynącego przez indukcyjność w chwili rozpoczęcia analizy stanów nieustalonych.

Parametry: strona 115.

Opis modelu: strona 115.

**M****TRANZYSTOR POŁOWY MOS**

Postać ogólna:

M..... \_dren \_bramka \_źródło \_podłoże \_nazwa\_modelu  
 +[L=\_var1] [W=\_var2] [AD=\_var3] [AS=\_var4] [PD=\_var5] [PS=\_var6]  
 +[NRD=\_var7] [NRS=\_var8] [NRG=\_var9] [NRB=\_var10]

Przykłady:

```
M1 14 2 13 0 PNOM L=25U W=12U
MAS 15 3 0 0 PSTRONG
MA2 0 2 100 100 NWEK L=33U W=12U AD=288P AS=288P PD=60U
+PS=60U NRD=14 NRS=24 NRG=10
```

Parametry, które można podać w linii deklaracji tranzystora MOS mają następujące znaczenie:

- L** Długość kanału tranzystora. Wartość domyślna wynosi 100[ $\mu\text{m}$ ].
- W** Szerokość kanału tranzystora. Wartość domyślna wynosi 100[ $\mu\text{m}$ ].

- AD** Pole powierzchni obszaru dyfuzji drenu. Służy do obliczania prądu nasycenia oraz pojemności złącza dren–podłoże.
- AS** Pole powierzchni obszaru dyfuzji źródła. Służy do obliczania prądu nasycenia oraz pojemności złącza źródło–podłoże.
- PD** Obwód obszaru dyfuzji drenu. Służy do obliczania pojemności części bocznej (zakrzywionej) złącza dren–podłoże.
- PS** Obwód obszaru dyfuzji źródła. Służy do obliczania pojemności części bocznej (zakrzywionej) złącza źródło–podłoże.
- NRD** Względna rezystancja obszaru drenu wyrażona liczbą kwadratów.
- NRS** Względna rezystancja obszaru źródła wyrażona liczbą kwadratów.
- NRG** Względna rezystancja obszaru bramki wyrażona liczbą kwadratów.
- NRB** Względna rezystancja obszaru podłoża wyrażona liczbą kwadratów.

Parametry: strony 165, 176, 184.

Opis modelu: strony 166, 174, 184.

## Q

## TRANZYSTOR BIPOLARNY

Postać ogólna:

Q..... \_kolektor \_baza \_emiter [\_podłoże] \_nazwa\_modelu [\_area]

Przykłady:

Q1 14 2 13 PNOM

Q13 15 3 0 1 NPNSTRONG 1.5

Liczba podana w polu **\_area** określa ile razy pole powierzchni zajmowane przez deklarowany przyrząd jest większe od pola powierzchni założonego w modelu. Parametry modelu **ISC** oraz **ISE** oznaczające prądy nasycenia dla prądów upływu złącza kolektor–baza i złącza emiter–baza mogą przyjąć wartości >1.0. W tym przypadku prądy nasycenia dla prądów upływu obliczane są jako iloczyn parametru **IS** (prąd nasycenia złącza) i odpowiednio parametru **ISC** oraz **ISE**.

Parametry: strony 141, 148.

Opis modelu: strona 139.

## R

## OPORNIK

Postać ogólna:

R..... n+ n- [\_nazwa\_modelu] \_oporność

Przykłady:

RLOAD 15 0 2K

R2 1 2 MOJ 2.0

Dodatni biegun opornika to **n+**, ujemny biegun to **n-**. Dodatni prąd płynie do bieguna dodatniego do bieguna ujemnego przez opornik. Wartość oporności podana w polu **\_oporność** może być dodatnia lub ujemna nie może być jednak równa zeru. Za pomocą modelu można opisać zależność oporności od temperatury.

Parametry: strona 112.

Opis modelu: strona 112.

## S KLUCZ STEROWANY NAPIĘCIEM

Postać ogólna:

S..... n+ n- nc+ nc- \_nazwa\_modelu

Przykłady:

S12 13 17 2 0 SMOD

SRESET 5 0 15 3 RELAY

Klucz sterowany napięciem jest rodzajem nieliniowego opornika wpiętego między węzły **n+** i **n-**. Wartość oporności zmienia się w sposób ciągły przy zmianach napięcia panującego między węzłami **nc+** i **nc-**.

Parametry: strona 78.

## T BEZSTRATNA LINIA DŁUGA

Postać ogólna:

T..... na+ na- nb+ nb- Z0=\_var1 [TD=\_var2] [F=\_var3] [NL=\_var4]

Przykłady:

T1 1 2 3 4 Z0=220OHM TD=115NS

T2 1 2 3 4 Z0=220OHM F=2.25MHz

T3 1 2 3 4 Z0=220OHM F=4.5MHz NL=0.5

Wejście linii dołączone jest do węzłów o numerach **na+** i **na-**. Wyjście linii dołączone jest do węzłów numerach **nb+** i **nb-**. Po słowie kluczowym **Z0** podaje się impedancję falową linii. Czas przelotu linii podaje się bezpośrednio po słowie kluczowym **TD** lub przez podanie częstotliwości fali (parametr **F**) i względnej długości linii **NL**. Jeżeli parametr **NL** nie zostanie podany to przyjmuje się, że **NL**=0.25, tzn. że linia jest ćwierćfalowa.

Patrz strona: 79.

## V NIEZALEŻNE ŹRÓDŁO NAPIĘCIA — SEM

Postać ogólna:

V..... n+ n- [DC \_war] [AC \_amp [\_faz]] [\_przebieg\_czasowy]

Przykłady:

VBIAS 12 0 2.3mV

VAC 2 3 AC 0.001

VACFAZ 2 3 AC .001 50

VPULSE 1 0 PULSE(0 1)

V3 26 77 DC 12 AC 1.2 56 SIN(1.2 12 200kHz)

Dodatni biegun źródła to **n+**, ujemny biegun źródła to **n-**. Dodatni prąd płynie od bieguna dodatniego do ujemnego przez źródło. Składową stałą napięcia źródła podaje się w polu **\_war** po słowie kluczowym **DC**. Po słowie kluczowym **AC** podaje się amplitudę (pole **\_amp**) i fazę (pole **\_faz**) składowej zmiennej źródła. Na końcu deklaracji można podać przebieg czasowy prądu źródła. Możliwe są następujące przebiegi:

**EXP** Ekspotencjalny — opis na stronie 76.

**PULSE** Przebieg prostokątny — opis na stronie 74.

**PWL** Odcinkowo–liniowy — opis na stronie 77.

**SFFM** Sinusoida o częstotliwości modulowanej sinusoidą — opis na stronie 77.  
**SIN** Sinusoidalny — opis na stronie 75.

Patrz strona: 9, 46, 73.

### **W** **KLUCZ STEROWANY PRĄDEM**

Postać ogólna:

W..... n+ n- \_nazwa\_źródła \_nazwa\_modelu

Przykłady:

W12 13 17 VC WMOD

WRESET 5 0 VRESET RELAY

Klucz sterowany prądem jest rodzajem nieliniowego opornika wpiętego między węzły **n+** i **n-**. Wartość oporności zmienia się w sposób ciągły przy zmianach prądu płynącego przez niezależne źródło napięcia, którego nazwa została podana w polu **\_nazwa\_źródła**.

Patrz strona: 78.

### **X** **PODOBWÓD**

Postać ogólna:

X..... <\_lista\_węzłów> \_nazwa\_podobwodu

Przykłady:

X12 100 101 200 201 DIFAMP

XBUFF 13 15 UNITAMP

Pseudoelement o nazwie zaczynającej się od litery „X” służy do włączenia podobwodu o nazwie podanej w polu **\_nazwa\_podobwodu** w strukturę obwodu. Na liście węzłów **\_lista\_węzłów** powinny się znaleźć numery węzłów obwodu, do których dołączone zostaną węzły podobwodu udostępniane na zewnątrz. Liczba węzłów umieszczonych na liście **\_lista\_węzłów** musi być taka sama jak liczba węzłów zadeklarowana w deklaracji .SUBCKT. Wywołania podobwodów mogą być zagnieżdżane.

Patrz strona: 68.